

## NAG Library Routine Document

### F07FAF (DPOSV)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

#### 1 Purpose

F07FAF (DPOSV) computes the solution to a real system of linear equations

$$AX = B,$$

where  $A$  is an  $n$  by  $n$  symmetric positive definite matrix and  $X$  and  $B$  are  $n$  by  $r$  matrices.

#### 2 Specification

```
SUBROUTINE F07FAF (UPLO, N, NRHS, A, LDA, B, LDB, INFO)
  INTEGER          N, NRHS, LDA, LDB, INFO
  REAL (KIND=nag_wp) A(LDA,*), B(LDB,*)
  CHARACTER(1)    UPLO
```

The routine may be called by its LAPACK name *dposv*.

#### 3 Description

F07FAF (DPOSV) uses the Cholesky decomposition to factor  $A$  as  $A = U^T U$  if UPLO = 'U' or  $A = LL^T$  if UPLO = 'L', where  $U$  is an upper triangular matrix and  $L$  is a lower triangular matrix. The factored form of  $A$  is then used to solve the system of equations  $AX = B$ .

#### 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

#### 5 Parameters

- 1: UPLO – CHARACTER(1) *Input*  
*On entry:* if UPLO = 'U', the upper triangle of  $A$  is stored.  
 If UPLO = 'L', the lower triangle of  $A$  is stored.  
*Constraint:* UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the number of linear equations, i.e., the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 3: NRHS – INTEGER *Input*  
*On entry:*  $r$ , the number of right-hand sides, i.e., the number of columns of the matrix  $B$ .  
*Constraint:* NRHS  $\geq 0$ .

- 4: A(LDA,\*) – REAL (KIND=nag\_wp) array Input/Output  
**Note:** the second dimension of the array A must be at least  $\max(1, N)$ .  
*On entry:* the  $n$  by  $n$  symmetric matrix  $A$ .  
 If UPLO = 'U', the upper triangular part of  $A$  must be stored and the elements of the array below the diagonal are not referenced.  
 If UPLO = 'L', the lower triangular part of  $A$  must be stored and the elements of the array above the diagonal are not referenced.  
*On exit:* if INFO = 0, the factor  $U$  or  $L$  from the Cholesky factorization  $A = U^T U$  or  $A = LL^T$ .
- 5: LDA – INTEGER Input  
*On entry:* the first dimension of the array A as declared in the (sub)program from which F07FAF (DPOSV) is called.  
*Constraint:*  $LDA \geq \max(1, N)$ .
- 6: B(LDB,\*) – REAL (KIND=nag\_wp) array Input/Output  
**Note:** the second dimension of the array B must be at least  $\max(1, NRHS)$ .  
*On entry:* the  $n$  by  $r$  right-hand side matrix  $B$ .  
*On exit:* if INFO = 0, the  $n$  by  $r$  solution matrix  $X$ .
- 7: LDB – INTEGER Input  
*On entry:* the first dimension of the array B as declared in the (sub)program from which F07FAF (DPOSV) is called.  
*Constraint:*  $LDB \geq \max(1, N)$ .
- 8: INFO – INTEGER Output  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

The leading minor of order  $\langle value \rangle$  of  $A$  is not positive definite, so the factorization could not be completed, and the solution has not been computed.

## 7 Accuracy

The computed solution for a single right-hand side,  $\hat{x}$ , satisfies an equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and  $\epsilon$  is the *machine precision*. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x} - x\|_1}{\|x\|_1} \leq \kappa(A) \frac{\|E\|_1}{\|A\|_1},$$

where  $\kappa(A) = \|A^{-1}\|_1 \|A\|_1$ , the condition number of  $A$  with respect to the solution of the linear equations. See Section 4.4 of Anderson *et al.* (1999) for further details.

F07FBF (DPOSVX) is a comprehensive LAPACK driver that returns forward and backward error bounds and an estimate of the condition number. Alternatively, F04BDF solves  $Ax = b$  and returns a forward error bound and condition estimate. F04BDF calls F07FAF (DPOSV) to solve the equations.

## 8 Parallelism and Performance

F07FAF (DPOSV) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F07FAF (DPOSV) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of floating-point operations is approximately  $\frac{1}{3}n^3 + 2n^2r$ , where  $r$  is the number of right-hand sides.

The complex analogue of this routine is F07FNF (ZPOSV).

## 10 Example

This example solves the equations

$$Ax = b,$$

where  $A$  is the symmetric positive definite matrix

$$A = \begin{pmatrix} 4.16 & -3.12 & 0.56 & -0.10 \\ -3.12 & 5.03 & -0.83 & 1.18 \\ 0.56 & -0.83 & 0.76 & 0.34 \\ -0.10 & 1.18 & 0.34 & 1.18 \end{pmatrix} \quad \text{and} \quad b = \begin{pmatrix} 8.70 \\ -13.35 \\ 1.89 \\ -4.14 \end{pmatrix}.$$

Details of the Cholesky factorization of  $A$  are also output.

### 10.1 Program Text

```

Program f07faf

!       F07FAF Example Program Text

!       Mark 25 Release. NAG Copyright 2014.

!       .. Use Statements ..
!       Use nag_library, Only: dposv, nag_wp, x04caf
!       .. Implicit None Statement ..
!       Implicit None
!       .. Parameters ..
!       Integer, Parameter          :: nin = 5, nout = 6
!       .. Local Scalars ..
!       Integer                    :: i, ifail, info, lda, n
!       .. Local Arrays ..
!       Real (Kind=nag_wp), Allocatable :: a(:,,:), b(:)
!       .. Executable Statements ..
!       Write (nout,*) 'F07FAF Example Program Results'
!       Write (nout,*)
!       Skip heading in data file
!       Read (nin,*)

```

```

      Read (nin,*) n
      lda = n
      Allocate (a(lda,n),b(n))

!      Read the upper triangular part of A from data file

      Read (nin,*)(a(i,i:n),i=1,n)

!      Read b from data file

      Read (nin,*) b(1:n)

!      Solve the equations Ax = b for x
!      The NAG name equivalent of dposv is f07faf
      Call dposv('Upper',n,1,a,lda,b,n,info)

      If (info==0) Then

!         Print solution

         Write (nout,*) 'Solution'
         Write (nout,99999) b(1:n)

!         Print details of factorization

         Write (nout,*)
         Flush (nout)

!         ifail: behaviour on error exit
!         =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
         ifail = 0
         Call x04caf('Upper','Non-unit diagonal',n,n,a,lda,'Cholesky factor U', &
            ifail)

      Else
         Write (nout,99998) 'The leading minor of order ', info, &
            ' is not positive definite'
      End If

99999 Format ((3X,7F11.4))
99998 Format (1X,A,I3,A)
      End Program f07faf

```

## 10.2 Program Data

```

F07FAF Example Program Data
  4                               :Value of N
  4.16  -3.12  0.56  -0.10
           5.03  -0.83  1.18
                0.76  0.34
                        1.18 :End of matrix A
  8.70 -13.35  1.89  -4.14 :End of vector b

```

## 10.3 Program Results

F07FAF Example Program Results

```

Solution
  1.0000   -1.0000    2.0000   -3.0000

Cholesky factor U
      1      2      3      4
  1  2.0396  -1.5297  0.2746  -0.0490
  2      1.6401  -0.2500  0.6737
  3      0.7887  0.6617
  4      0.5347

```

---